

Paper 408-2013

## Destination Known: Programmatically Controlling Your Output in SAS® Enterprise Guide®

Aaron Hill, MDRC, New York, NY

### ABSTRACT

In a SAS® Enterprise Guide® project with multiple reports and graphics, you can organize output by selectively sending content to different ODS destinations embedded within the project. For example, within a single program you could embed tables in RTF format, graphics in PDF, and other output in Text. The SAS syntax is simple and gives you programmatic control over all output and destinations. The result: a well-organized project with all results in their preferred format.

### INTRODUCTION

SAS programs, logs, and results are embedded within an Enterprise Guide Project File (\*.egg) by default. Results may be output as SAS Report, HTML, PDF, RTF, and Text—allowing for any combination of these formats. It's easy to choose result formats to embed in the project in the global Options window (Tools -> Options -> Results General, under Result Formats) or in the Program Properties (under Results).

However, you may not always want to send all results to all open destinations, instead preferring to selectively send output to specific destinations. You also may want the output to be the same when a project is shared among multiple users, regardless of individual users' settings. This paper demonstrates how to programmatically control output in Enterprise Guide. In the examples that follow, SAS code is used to send select tabular output to Text, select graphics to SAS Report, select tables to HTML (with each in a different output window), and two similar reports to PDF with different specifications—all embedded in the project.

### SELECTIVELY EMBEDDING OUTPUT IN A PROJECT

Based on Results options selected in Tools or Program Properties, Enterprise Guide will include wrapper code that opens each of the specified locations. Because the intention is to override these settings and explicitly specify the ODS destinations programmatically, a program should start by using ODS CLOSE statements to close each of the destinations opened in the wrapper code, or by using:

```
ods _all_ close;
```

Note: The wrapper code is generated every time SAS is run, including when running select pieces of code within a program. Because of this, use ODS CLOSE statements at the top of any selection of code to be run. In some cases, ODS \_ALL\_ CLOSE will suppress automatically generated output.

### OPENING A DESTINATION

Control of the Text output is easiest to program and the Listing is automatically embedded in the project. Open and close the Listing destination as needed. After closing the Listing destination and re-opening it later, new output is appended.

```
ods listing;  
proc contents data=sashelp.shoes varnum;  
title "Histogram of Returns";  
run;  
ods listing close;
```

For all other ODS destinations, embedding output in a project involves an additional step. When writing an ODS statement that directs results to an external location, Enterprise Guide writes the output file and by default creates a link to the file rather than redirect it to the project. For example, an ODS RTF statement with a FILE= option to send output to 'C:\EG\MYREPORT.RTF', the MYREPORT.RTF file would be written to C:\EG and the project would contain a link to this external file (not embedded in the project). However, the focus of this paper is programmatically embedding all output in a project. This requires writing ODS statements that direct the results to the WORK directory; because there is no external destination, the temporary ODS output files will be redirected to the project. Use a filename statement to reference a new output file and designate it as temporary:

```
filename egsr temp;
```

The filename **egsr** could be any valid name. This output file will be used below for the SAS Report destination, so **egsr** is a naming convention consistent with its intended use. The TEMP option specifies that it is a temporary file in the WORK directory. After designating the temporary output file, refer to this file in an ODS statement. Beyond this, write ODS statements as you normally would.

```
ods tagsets.sasreport13 file=egsr gpath=&sasworklocation;
```

The SAS Report destination was opened with ODS TAGSETS.SASREPORT13 (“SAS Report” is actually an ODS MARKUP template; the latest SAS Report template is SASREPORT13). The previously specified **egsr** file was referenced as the output location with FILE=EGSR. Finally, the location of all graphics output was specified with GPATH=&SASWORKLOCATION;. The GPATH option is necessary in SAS Report, HTML, and Listing destinations when the output includes graphics. The macro variable **sasworklocation** is automatically assigned by Enterprise Guide; it references the default WORK directory graphics location.

Now that all other ODS destinations are closed and SAS Report is open and writing to **egsr**, output will be embedded in the project:

```
ods noproctitle; ■ the ODS NOPROCTITLE statement suppresses procedure titles
```

```
proc univariate data=sashelp.shoes noprint;
  histogram Returns;
  title "Histogram of Returns";
run;
```




Figure 1. Partial SAS Report output embedded in project

## USING ODS EXCLUDE AND SELECT TO CONTROL OUTPUT

At this point, the SAS Report ODS destination is still open. If it were closed, output could no longer be appended to the **egsr** file later in the program. In fact, closing and reopening the destination with output for the **egsr** file would overwrite previous output. To leave the destination open while maintaining control of whether output is written, use ODS EXCLUDE and ODS SELECT. Suppose that the code following the PROC UNIVARIATE generates output not intended to be written to SAS Report. However, later in the program more output will be written to the SAS Report destination. Rather than close the destination, temporarily suppress all output from this destination using ODS EXCLUDE ALL. Later, when ready to resume writing to SAS Report, use ODS SELECT ALL. For example:

```
ods tagsets.sasreport13 exclude all; ■ suppresses all output from SAS Report
[output-generating code that is sent to other ODS destination(s)]
ods tagsets.sasreport13 select all; ■ resumes writing output to SAS Report
[output-generating code is appended to the open SAS Report destination]
ods tagsets.sasreport13 close; ■ no further output to be written to SAS Report; close destination
```

## CREATING MULTIPLE ODS OUTPUT FILES IN THE SAME DESTINATION

Suppose a report should contain a PROC PRINT for each of two regions in HTML format embedded in the project. It may be desirable to have each region's PRINT displayed in its own window rather than a single window of sequential PRINTs. To do this, open and close multiple instances of the same destination, each with a unique fileref. The following example shows two HTML reports, each sent to a separate output window:

```
filename Africa temp; ■ the first HTML output file is named "Africa"

ods html file=Africa gpath=&sasworklocation ;
proc print data=sashelp.shoes noobs;
var Product Subsidiary Stores Sales Inventory Returns;
where region = "Africa";
title "Records in Africa Region";
run;
ods html close; ■ close this HTML output file

filename Asia temp; ■ the second HTML output file is named "Asia"

ods html file=Asia gpath=&sasworklocation ;
proc print data=sashelp.shoes noobs;
var Product Subsidiary Stores Sales Inventory Returns;
where region = "Asia";
title "Records in Asia Region";
run;
ods html close; ■ close this HTML output file
```




Figure 2. Two HTML output files viewable by selecting a window in the drop-down menu

## USING ID= TO REFERENCE MULTIPLE OUTPUT FILES IN A SINGLE DESTINATION

When writing multiple files to the same destination, the ID= option identifies each with a name so that specifications can be made for each file. In this example, two PDF files are being written. The first PDF file is a summary report, including only the CONTENTS of SASHELP.SHOES (but suppressing the 'Engine/Host Dependent Information' table from the output). The second PDF report contains the full CONTENTS report, as well as PROC FREQ and PROC MEANS reports. Adding the ID= option (in parentheses, immediately after the ODS statement keyword) designates a name for later reference:

```
filename summary temp;
ods pdf(id=summary) file=summary; ■ ID= with the name "summary"

filename full temp;
ods pdf(id=full) file=full; ■ ID= with the name "full"
```

Now specific instructions can be provided for each PDF output file by referencing the ID name. Both reports contain the CONTENTS of SASHELP.SHOES, but the summary report will exclude the 'Engine/Host Dependent Information' table from the default CONTENTS output. Refer to documentation on the ODS TRACE statement for more information on identifying and referencing specific output objects.

```
ods pdf(id=summary) exclude EngineHost; ■ Exclude EngineHost output object from SUMMARY

proc contents data=sashelp.shoes varnum;
title "Report: Contents of SHOES Data";
run;
```

The summary report ends here, so exclude all of the following ODS output from this file:

```
ods pdf(id=summary) exclude all; ■ Exclude all further output from SUMMARY

proc freq data=sashelp.shoes;
tables _character_;
title "Full Report: Shoes Data";
run;

proc means data=sashelp.shoes;
var _numeric_;
run;

ods pdf(id=summary) close;
ods pdf(id=full) close;
```

The result is two PDF reports embedded in the project, each in a separate window. The Summary Report contains a condensed version of CONTENTS and the Full Report contains the full CONTENTS as well as FREQ and MEANS output. The reports are displayed in separate output windows:




Figure 3. Summary Report

Figure 4. Full Report

## CONCLUSIONS

Using SAS code to control the ODS output embedded in an Enterprise Guide project is a great way to organize output. The syntax is simple and offers programmatic control over all output and destinations. The result: a well-organized project with all results in their preferred format. An appendix includes a comprehensive example of the code and output described in this paper.

- Always start by closing all ODS destinations that were opened in wrapper code (based on user settings).
- Create new temporary file references with FILENAME statements including the TEMP option and reference these output files in ODS statements. They will automatically be embedded in the project.
- After destinations have been opened, use ODS EXCLUDE and ODS SELECT to selectively suppress and include output. Close a destination only when no further output should be sent there.
- Create multiple output files for the same destination to create a separate output window for each.
- Use the ID= option to name and reference multiple instances of output to the same destination.

## REFERENCES

SAS Institute Inc. *Base SAS® 9.3 Procedures Guide*. Cary, NC: SAS Institute Inc., 2011.

## ACKNOWLEDGMENTS

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The author thanks Nathaniel Roth for his work at MDRC on this topic, a version of which he presented at the MDRC SAS Users Group.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Aaron Hill

Twitter: @Research4Truth

Email: [aaron.hill@mdrc.org](mailto:aaron.hill@mdrc.org) or [aaron.hill@me.com](mailto:aaron.hill@me.com)

MDRC

16 East 34<sup>th</sup> Street, 19<sup>th</sup> Floor

New York, NY 10016

Work Phone: (212) 340-8898

Web: [www.mdrc.org](http://www.mdrc.org)

## APPENDIX

This program sends graphical output to SAS Report, ten PRINT reports to HTML in separate output windows, two PDF reports using ID= option to specify differences between overlapping content, and all other output to Listing. Following the SAS code are figures showing how results are organized in the Enterprise Guide project.

```
ods _all_ close;

*send to Listing | See Appendix Figure 4;

ods listing;
proc contents data=sashelp.shoes varnum;
title "Contents of SHOES data";
run;
ods listing close;

*send to SAS Report (leave destination open for further use) | See Appendix Fig. 1;

filename egsr temp;
ods tagsets.sasreport13 file=egsr gpath=&sasworklocation;

ods noproctitle;

proc univariate data=sashelp.shoes noprint;
histogram Returns;
title "Histogram of Returns";
run;

ods tagsets.sasreport13 exclude all;
ods proctitle;

*send 10 PRINT reports to HTML, each with own tab | See Appendix Figure 2;

%macro REGREPORT(tempfile=, region=);
filename &tempfile temp;
ods html file= &tempfile gpath=&sasworklocation ;
proc print data=sashelp.shoes noobs;
var Product Subsidiary Stores Sales Inventory Returns;
where region = "&region";
title "Records in &region Region";
run;
ods html close;
%mend REGREPORT;

%regreport(tempfile= AFRICA,    region=Africa)
%regreport(tempfile= ASIA,    region=Asia)
%regreport(tempfile= CANADA,  region=Canada)
%regreport(tempfile= CAMERICA, region=Central America/Caribbean)
%regreport(tempfile= EASTEURO, region=Eastern Europe)
%regreport(tempfile= MIDEAST,  region=Middle East)
%regreport(tempfile= PACIFIC,  region=Pacific)
%regreport(tempfile= SAMERICA, region=South America)
%regreport(tempfile= USA,      region=United States)
%regreport(tempfile= WESTEUR,  region=Western Europe)

*append to listing output | See Appendix Figure 4;

ods listing;
proc freq data=sashelp.shoes;
tables _character_ ;
title "Frequencies of character variables in SHOES";
run;
```

```
proc means data=sashelp.shoes;
var _numeric_;
title "Means of numeric variables in SHOES";
run;
ods listing close;

*append to SAS Report output | See Appendix Figure 1;

ods tagsets.sasreport13 select all;
ods noproctitle;
title;

proc sgscatter data=sashelp.shoes;
plot returns * sales;
title "Scatterplot of Returns x Sales";
run;

ods tagsets.sasreport13 close;
ods proctitle;

*send two report versions to PDF | See Appendix Figure 3:
  1) summary report (CONTENTS only, without EngineHost output object)
  2) full report (CONTENTS, FREQ, and MEANS);

filename summary temp;
ods pdf(id=summary) file=summary;

filename full temp;
ods pdf(id=full) file=full;


* Exclude Engine/Host Dependent Information table from the summary report;

ods pdf(id=summary) exclude EngineHost;
proc contents data=sashelp.shoes varnum;
title "Report: Contents of SHOES Data";
run;

* Exclude the following output from the Summary Report;

ods pdf(id=summary) exclude all;
proc freq data=sashelp.shoes;
tables _character_;
title "Full Report: Shoes Data";
run;

proc means data=sashelp.shoes;
var _numeric_;
run;
ods _all_ close;
```




Appendix Figure 1. Histogram and scatterplot to SAS Report


The screenshot shows the SAS Enterprise Guide interface. A red box highlights the "Results - HTML (10)" window in the taskbar. On the left, a tree view shows a folder named "SAMERICA" which is expanded to show a list of regions: SAMERICA, PACIFIC, USA, MIDEAST, WESTEUR, CANADA, ASIA, AFRICA, CAMERICA, and EASTEURO. The main window displays a report titled "South America Region" with a table containing the following data:

	Subsidiary	Stores	Sales	Inventory	Returns
	Bogota	19	\$15,312	\$35,805	\$1,229
	Bogota	2	\$20,835	\$43,155	\$1,136
	Bogota	4	\$37,187	\$62,770	\$1,424
Sandal	Bogota	16	\$22,712	\$62,856	\$1,292
Slipper	Bogota	15	\$33,723	\$88,411	\$2,033
Sport Shoe	Bogota	19	\$4,729	\$11,597	\$333
Women's Dress	Bogota	14	\$71,736	\$117,761	\$2,646
Boot	Buenos Aires	16	\$17,027	\$39,291	\$1,073
Men's Casual	Buenos Aires	1	\$16,019	\$15,889	\$498
Men's Dress	Buenos Aires	1	\$14,437	\$29,932	\$354
Sandal	Buenos Aires	21	\$18,765	\$75,358	\$604
Slipper	Buenos Aires	8	\$11,759	\$23,822	\$939
Sport Shoe	Buenos Aires	14	\$3,545	\$6,103	\$229
Women's Casual	Buenos Aires	1	\$15,470	\$19,479	\$382

Appendix Figure 2. Ten proc print reports to HTML, each in a separate output window



Appendix Figure 3. Two PDF reports using ID=



Appendix Figure 4. contents, freq, and means to Listing